379 No/

MSC-IN-64-ED-6

MSC INTERNAL TECHNICAL NOTE

THE CONCEPT OF GENERALIZED INVERSION OF ARBITRARY COMPLEX MATRICES

BY

Henry P. Decell, Jr.

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

MANNED SPACECRAFT CENTER

HOUSTON, TEXAS
July 16, 1964

MSC-IN-64-ED-6

MSC INTERNAL TECHNICAL NOTE

THE CONCEPT OF GENERALIZED INVERSION OF ARBITRARY COMPLEX MATRICES

Prepared by: Henry P. Decell, Jr.
Henry P. Decell, Jr.

Theory and Analysis Office
Computation and Analysis Division

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

MANNED SPACECRAFT CENTER

HOUSTON, TEXAS
July 16, 1964

TABLE OF CONTENTS

Section	Page
LIST OF SYMBOLS	ii
INTRODUCTION	1
DEFINITIONS AND EQUIVALENT FORMS	2
THE EXPLICIT FORM	9
APPLICATION TO LEAST SQUARES APPROXIMATION	12
SUBROUTINE GENINV	14
REFERENCES	23

LIST OF SYMBOLS

1. Capital letters denote matrices unless otherwise stated.
2. Lower case letters denote column vectors unless otherwise stated or clear from context.
3. A^* denotes the matrix conjugate transpose of A .
4. A^{-1} denotes matrix inverse for nonsingular A .
5. A^+ denotes the generalized inverse of A .
6. H will denote a hermitian idempotent matrix (h.i.) i.e. a matrix such that $H^* = H$ and $HH = H$.
7. $R(A)$ denotes the range space of A i.e., the collection of all images of column vectors under the transformation A .
8. $P_{R(A)}$ will denote the orthogonal projection on the range of A i.e. a hermitian idempotent leaving $R(A)$ fixed.
9. E^m will denote m - dimensional euclidean space.
10. $\mathrm{diag}(a_1, a_2, \ldots, a_n)$ denotes a diagonal matrix.

INTRODUCTION

The primary concern of this paper is to investigate the problem of inversion of singular or non-square matrices. In this connection, a new algorithm for computing the generalized inverse of an arbitrary complex matrix is given. For a non-singular matrix the algorithm gives the ordinary inverse of the matrix.

The paper is divided into several sections. The first two sections give a definition-theorem exposé of the known results in the literature. The following sections give a new explicit form together with an algorithm for computing the new explicit form. An application to least squares approximation is given that can easily be realized in trajectory analysis problems. Finally, a computer program for computing the generalized inverse of a matrix is given utilizing the algorithm mentioned in the latter paragraph.

DEFINITIONS AND EQUIVALENT FORMS

A. Bjerhammer [2][1], E. H. Moore [10], and R. Penrose [11] independently generalized the concept of matrix inversion to include arbitrary complex matrices. The generalized inverse of a singular or non-square matrix possesses properties which make it a central concept in matrix theory.

We will give a definition-theorem exposé, inserting where applicable, references and special problems. The following fundamental theorem due to Penrose [11] will be stated without proof.

THEOREM 1. The four equations

(1) $AXA = A$

(2) $XAX = X$

(3) $(AX)^* = AX$

(4) $(XA)^* = XA$

have a unique solution X for each complex matrix A.

Definition 1. The solution X in THEOREM 1. will be denoted $X = A^+$ and called the generalized inverse of A.

The following theorem gives an equivalent form of A^+.

THEOREM 2. For any $m \times m$ matrix A over the complex field, $X = A^+$ is the unique solution to the equations

---

[1] Numbers in brackets refer to correspondingly numbered papers in the references.

$$AX = P_{R(A)}$$

and

$$XA = P_{R(X)}$$

where $R(A)$ is the range space of A in E^m and $P_{R(A)}$ is the orthogonal projection on $R(A)$.

Proof: THEOREM 1. implies that AX is a hermitian idempotent (see list of symbols) leaving A fixed i.e., $(AX)A = A$. Hence AX must be a projection. We may conclude the same about XA .

We proceed to give properties of the generalized inverse and possible computing schemes.

THEOREM 3. Let A be an arbitrary complex matrix. Then, for scalar $\lambda \neq 0$ and unitary U and V

(a) $A^+(A^+)^*A^* = A^+ = A^*(A^+)^*A^+$

(b) $A^+AA^* = A^* = A^*AA^+$

(c) $(A^+)^+ = A$

(d) $(A^*)^+ = (A^+)^*$

(e) $A^+ = A^{-1}$ for nonsingular A .

(f) $(\lambda A)^+ = \frac{1}{\lambda}A^+$

(g) $(A^*A)^+ = A^+(A^+)^*$

(h) $(UAV)^+ = V^{-1}A^+U^{-1}$

(i) $A = \sum A_i$ and $A_i^* A_j = 0$
$A_j^* A_i = 0$ for $i \neq j$

imply
$$A^+ = \sum A_i^+$$

(j) If A is normal (i.e. $A^*A = AA^*$)
then, $A^+A = AA^+$ and $(A^n)^+ = (A^+)^n$

(k) A, A^*A, A^+ and A^+A all have rank equal to trace (A^+A).

(l) $A^+ = (A^*A)^+A^*$

We note that (l) reduces the problem of computing A^+ to that of computing the generalized inverse of a hermitian matrix A^*A. Moreover, such a matrix can always be diagonalized by a unitary transformation i.e.,

$$D = U(A^*A)V = \text{diag } (a_1, \ldots, a_n)$$

Now (f) and (h) imply

$$(A^*A)^+ = VD^+U = V \text{ diag } (\frac{1}{a_1}, \ldots, \frac{1}{a_n})U$$

We tacitly assume that if $a_i = 0$ then the corresponding term in diag $(\frac{1}{a_1}, \ldots, \frac{1}{a_n})$ is zero. It is not usually an easy task to determine the unitary transformations U and V. Methods for computing the generalized inverse have geen given by various authors [2], [3], [7], [8], [12].

The following is a theorem of major importance characterizing all solutions of the matrix equations $AXB = C$ which have some solution X.

THEOREM 4. For the matrix equation $AXB = C$ to have a solution, a necessary and sufficient condition is:

$$AA^+CB^+B = C$$

in which case the general solution is:

$$X = A^+CB^+ + Y - A^+AYBB^+,$$

where Y is arbitrary (to within the limits of being consistent with dimension in the indicated multiplications) [11].

Proof: Suppose X satisfies $AXB = C$. Then,

$$C = AXB = AA^+AXBB^+B = AA^+CB^+B$$

Conversely, if $C = AA^+CB^+B$ then A^+CB^+ is a particular solution. Clearly, for the general solution we must solve $AXB = 0$. Any expression of the form

$$X = Y - A^+AYBB^+$$

is such a solution. Moreover, if $AXB = 0$ then,

$$X = X - A^+AXBB^+$$

We note that the only property required of A^+ and B^+ in the theorem is $AA^+A = A$, $BB^+B = B$.

COROLLARY 1. The general solution to the vector equation

$$Px = c \qquad \text{is}$$

$$x = P^+c + (I - P^+P)y$$

where y is arbitrary, provided a solution exists.

COROLLARY 2. A necessary and sufficient condition for the equations

$$AX = C$$

and

$$XB = D$$

to have a common solution is that each have a solution and $AD = CB$ [4].

Proof: If $AX = C$ and $XB = D$ have a common solution then clearly each has a solution and

$$AXB = CB$$
$$AXB = AD$$

so that

$$CB = AD$$

In order to obtain the sufficiency we set

$$X = A^{+}C + DB^{+} - A^{+}ADB^{+}$$

which is a solution if $AD = CB$, $AA^{+}C = C$, and $DB^{+}B = D$.

THEOREM 5. We have:

(1) $A^{+}A$, AA^{+}, $I-A^{+}A$, and $I - AA^{+}$ are h.i. (see list of symbols)

(2) H is h.i. implies $H^{+} = H$

Proof: The proof requires a straightforward application of THEOREM 1.

In general, the reversal rule (i.e., $(AB)^+ = B^+A^+$ as in the case of the standard inverse) does not hold. R. Cline [5] recently obtained the following result.

THEOREM 6. Let A and B be matrices with the product AB defined. Then,

$$(AB)^+ = B_1^+A_1^+$$

where $$AB = A_1B_1$$

and $$B_1 = A^+AB$$
$$A_1 = AB_1B_1^+$$

THE EXPLICIT FORM

Utilizing the properties of A^+ in the preceeding sections, we develop an explicit form which gives rise to an algorithm for computing the generalized inverse of an arbitrary complex matrix [7].

THEOREM 8. For any matrix A, $A^+ = WAY$, where W and Y are any solutions of

$$(1) \quad WAA^* = A^*$$

and

$$(2) \quad A^*AY = A^*$$

Proof: Equations (1) and (2) indeed have a solution $W = Y = A^+$. Moreover, if W and Y are any solutions we have

$$AWAA^* = AA^* \text{ and } A^*AYA = A^*A$$

so that

$$AWA = A \text{ and } AYA = A$$

(Note: $BAA^* = CAA^*$ implies $BA = CA$).

In addition,

$$WAA^*W^* = A^*W^* \text{ and } Y^*A^*AY = Y^*A^*$$

imply

$$(WA)^* = WA \text{ and } (AY)^* = AY$$

If we let $X = WAY$, X satisfies the four equations of THEOREM 1., so that $A^+ = X = WAY$.

COROLLARY 2. For any matrix A, $A^+ = A^*S_1AS_2A^*$ where S_1 and S_2 are, respectively, any solutions of

$$(AA^*)S_1(AA^*) = (AA^*)$$

and

$$(A^*A)S_2(A^*A) = (A^*A)$$

Proof: According to THEOREM 3. we have that $W = A^*S_1$ and $Y = S_2A^*$ are solutions of equations (1) and (2) of THEOREM 8. provided

$$(AA^*)S_1(AA^*) = (AA^*)$$

and

$$(A^*A)S_2(A^*A) = (A^*A)$$

The corollary follows.

THEOREM 9. If B is a matrix and there exist nonsingular matrices P and Q such that $PBQ = E$ is an idempotent then $\bar{B} = QEP$ is a solution of $BXB = B$.

Proof: If P, Q, and E satisfy the hypothesis of the theorem then $B = P^{-1}EQ^{-1}$ and

$$B\bar{B}B = (P^{-1}EQ^{-1})QEP(P^{-1}EQ^{-1}) = P^{-1}EQ^{-1} = B$$

COROLLARY 2. and THEOREM 9. suggest an algorithm for computing the generalized inverse of a complex matrix F. Consider the equation $F^+ = (F^*F)^+F^*$ [6], which reduces the problem of finding F^+ to that of finding the generalized inverse of the hermitian matrix $F^*F = C$. Since

$(C^2)^* = C^2$, there exist nonsingular matrices P and Q (products of elementary matrices obtained by simple elimination) such that

$$PC^2Q = \begin{pmatrix} I_r & Z \\ Z & Z \end{pmatrix} = I_o$$

where I_r is a rank r identity matrix and the Z are zero matrices. We set $C = A$ in COROLLARY 2., so that $A^*A = AA^* = C^*C = CC^* = C^2$. According to THEOREM 9. choose solutions $S_1 = S_2 = QI_oP$ so that $C^+ = (CS_1)^2C$, $(F^*F)^+ = C^+$, and finally,

$$F^+ = C^+F^*$$

Computing programs for calculating S_1 and S_2 are now in existence (e.g., STORM, Statistically Oriented Matrix Program, IBM). In general, these programs only compute some solution of the equation $AXA = A$, usually different from A^+. These results allow one to construct a solution to all four Penrose equations (THEOREM 1.), given only a solution of the first, namely, $AXA = A$.

APPLICATION TO LEAST SQUARES APPROXIMATION

We will now state an application that can be realized in trajectory analysis problems. For the sake of simplicity we will not consider weighting and only mention that weighting introduces no difficulty.

The vector equation $Ax = b$ does not, in general, have a solution x. However, all candidates for a least squares solution (i.e., a solution vector x minimizing $(Ax - b)^*(Ax - b)$ must be solutions of the normal equations

$$A^*Ax = A^*b \quad [8]$$

THEOREM 7. Let A be any matrix $(m \times n)$ and b be any vector $(m \times 1)$. The equation

$$A^*Ax = A^*b$$

always has a solution and hence a general solution given by:

$$\begin{aligned} X &= (A^*A)^+A^*b + (I - (A^*A)^+A^*A)y \\ &= A^+b + (I - A^+A)y \end{aligned}$$

Moreover, if A^*A is non-singular then the solution is

$$x = A^+b$$

and is unique.

Proof: We will first show that

$$(1) \quad A^*Ax = A^*b$$

has $\underline{\underline{a}}$ solution. Consider the vector:

$$x = A^{+}b$$

Since THEOREM 3.(b) implies $A^{*}A(A^{+}b) = A^{*}b$ we have that $x = A^{+}b$ is indeed a solution of (1). The existence of this solution together with COROLLARY 1. implies that the general solution to (1) is:

$$(2) \qquad x = (A^{*}A)^{+}A^{*}b + (I - (A^{*}A)^{+}A^{*}A)y$$

Using THEOREM 3.(1) we see that

$$x = A^{+}b + (I - A^{+}A)y.$$

Finally if $A^{*}A$ is non-singular then

$$\begin{aligned} x &= (A^{*}A)^{+}A^{+}b + (I - I)y \\ &= A^{+}b \end{aligned}$$

and (1) has a unique solution.

In summary, we know that if x is a least squares solution of $Ax = b$, then x must satisfy

$$A^{*}Ax = A^{*}b$$

All solutions of this equation are given by $x = A^{+}b + (I - A^{+}A)y$. Any vector of the form

$$x = A^{+}b + (I - A^{+}A)y$$

is a "candidate" for a least squares solution and this form describes the "class of all candidates."

COROLLARY 3. Every solution of $A^*Ax = A^*b$ minimizes $Q = (Ax - b)^*(Ax - b)$ provided Q has a minimum.

Proof: We know that any vector at which Q is minimum is of the form

$$x = A^+b + (I - A^+A)y$$

If Q has a minimum let

$$x_1 = A^+b + (I - A^+A)\ y_2$$

be any other solution. We will show that

$$(Ax_1 - b)^*(Ax_1 - b) = (Ax_2 - b)^*(Ax_2 - b)$$

To do this we examine Ax_1 and Ax_2 using THEOREM 1.

$$\begin{aligned} Ax_1 &= A(A^+b + (I - A^+A)y_1) \\ &= AA^+b + (A - AA^+A)y_1 \\ &= AA^+b + (I-I)y_1 \\ &= AA^+b \end{aligned}$$

Similarly $Ax_2 = AA^+b$
so that

$$(Ax_1 - b)^*(Ax_1 - b) = (Ax_2 - b)^*(Ax_2 - b)$$

that is, every vector of the form

$$x = A^+b + (I - A^+A)y$$

yields the same minimum value of Q .

SUBROUTINE GENINV

GENINV is a FORTRAN IV subroutine, written by L.F. Guseman, Jr., Theory and Analysis Office, which is used to compute the generalized inverse of an mxn matrix A . All computations are done in double-precision floating point arithmetic. The subroutine is based on the algorithm suggested by the explicit form.

CALLING SEQUENCE

CALL GENINV (A, AP, M, N, L. E),

where,

A — is a double-dimensioned, double-precision array containing the original matrix. A is dimensioned A(25, 25).

AP — is a double-dimensioned, double-precision array where the generalized inverse of A will be computed. AP is dimensioned AP(25, 25).

M — is the number of rows in the original matrix.

N — is the number of columns in the original matrix.

L — is twice N.

E — is some small number for near-zero divisor test.

METHOD

Given A (PRINT A)

Compute:

$C = A^*A$ (PRINT C)

$C^2 = CC$ (PRINT C^2)

Find non-singular matrices E and P such that

$$EC^2P = \begin{pmatrix} I_r & Z \\ Z & Z \end{pmatrix} = I_o$$

PRINT E, P, I_o

(A form of Gaussian elimination with pivoting employed)

Compute:

$$R = PI_oE \qquad \text{(PRINT R)}$$

then

$$C^+ = CRCRC \qquad \text{(PRINT } C^+\text{)}$$

also

$$A^+ = C^+A^* \qquad \text{(PRINT } A^+\text{)}$$

Remarks

The program uses two double-precision arrays CSQ(50, 50) and B(25, 25) for internal manipulation. The subroutine leaves the original matrix A intact.

Results are printed after each step as indicated.

```
      SUBROUTINE GENINV(A,AP,M,N,L,E)
      DIMENSION A(25,25),AP(25,25),CSQ(50,50),B(25,25)
      DOUBLE PRECISION CSQ,B,BIGA,X,DABS,A,AP
C*
C**   THIS SUBROUTINE COMPUTES THE GENERALIZED INVERSE OF A MATRIX
C**** ALGORITHM BY H. P. DECELL,
C**   CALLING SEQUENCE
C**
C**     CALL GENINV(A,AP,M,N,L,E)
C**
C**   A(M,N) - LOCATION OF ORIGINAL MATRIX
C**   AP(N,M) - LOCATION OF COMPUTED GENERALIZED INVERSE
C**   M - NUMBER OF ROWS IN ORIGINAL MATRIX
C**   N - NUMBER OF COLUMNS IN ORIGINAL MATRIX
C**   L - 2*N
C**   E - SMALL POSITIVE NUMBER FOR NEAR-ZERO DIVISOR TEST
C*
C     INITIALIZATION
      NP1=N+1
      WRITE(6,100)
  100 FORMAT(1H1,13H THE MATRIX A//)
      DO 50 I=1,M
      WRITE(6,200)(A(I,J),J=1,N)
   50 CONTINUE
C                          *
C     COMPUTATION OF C=A A
      DO 1 I=1,N
      I1=I+N
      DO 1 J=1,N
      J1=J+N
      CSQ(I1,J1)=0.0D0
      DO 1 K=1,M
      CSQ(I1,J1)=CSQ(I1,J1)+A(K,I)*A(K,J)
    1 CONTINUE
      WRITE(6,101)
  101 FORMAT(1H0,13H THE MATRIX C//)
      DO 51 I=NP1,L
      WRITE(6,200)(CSQ(I,J),J=NP1,L)
   51 CONTINUE
  200 FORMAT(1H ,6D21.12)
C                    2
C     COMPUTATION OF C =CC
      DO 2 I=1,N
      I1=I+N
      DO 2 J=1,N
      J1=J+N
      CSQ(I,J)=0.0D0
      DO 2 K=1,N
```

```
      K1=K+N
      CSQ(I,J)=CSQ(I,J)+CSQ(I1,K1)*CSQ(K1,J1)
    2 CONTINUE
      WRITE(6,107)
  107 FORMAT(1H0,19H THE MATRIX CSQUARE//)
      DO 510 I=1,N
      WRITE(6,200)(CSQ(I,J),J=1,N)
  510 CONTINUE
C
C     BUILD AUGMENTED MATRICES
      DO 5 I=1,N
      DO 5 J=NP1,L
      IF((J-N)-I)3,4,3
    3 CSQ(I,J)=0.0D0
      GO TO 5
    4 CSQ(I,J)=1.0D0
    5 CONTINUE
      DO 8 J=1,N
      DO 8 I=NP1,L
      IF((I-N)-J)6,7,6
    6 CSQ(I,J)=0.0D0
      GO TO 8
    7 CSQ(I,J)=1.0D0
    8 CONTINUE
C
C     COMPUTATION OF I
C                     0
C
C     A FORM OF GAUSSIAN ELIMINATION IS EMPLOYED
      DO 22 K=1,N
      KP1=K+1
      IR=K
      JC=K
      BIGA=DABS(CSQ(K,K))
      DO 10 I=K,N
      DO 10 J=K,N
      IF(BIGA-DABS(CSQ(I,J)))9,10,10
    9 IR=I
      JC=J
      BIGA=DABS(CSQ(I,J))
   10 CONTINUE
      IF(BIGA-E)23,23,11
C
C     EXCHANGE ROWS
   11 IF(IR-K)12,14,12
   12 DO 13 J=1,L
      X=CSQ(IR,J)
      CSQ(IR,J)=CSQ(K,J)
      CSQ(K,J)=X
   13 CONTINUE
C
C     EXCHANGE COLUMNS
   14 IF(JC-K)15,17,15
   15 DO 16 I=1,L
      X=CSQ(I,JC)
      CSQ(I,JC)=CSQ(I,K)
```

```
      CSQ(I,K)=X
   16 CONTINUE
C
C     DIVIDE ROW K BY CSQ(K,K)
   17 X=CSQ(K,K)
      DO 18 J=K,L
      CSQ(K,J)=CSQ(K,J)/X
   18 CONTINUE
      IF(K-N)19,22,22
C
C     ZERO COLUMN K BELOW THE DIAGONAL
   19 DO 20 I=KP1,N
      X=CSQ(I,K)
      DO 20 J=K,L
      CSQ(I,J)=CSQ(I,J)-X*CSQ(K,J)
   20 CONTINUE
C
C     ZERO ROW K TO THE RIGHT OF THE DIAGONAL
      DO 21 J=KP1,N
      X=CSQ(K,J)
      DO 21 I=K,L
      CSQ(I,J)=CSQ(I,J)-X*CSQ(I,K)
   21 CONTINUE
   22 CONTINUE
   23 CONTINUE
      WRITE(6,102)
  102 FORMAT(1H0,17H THE MATRIX IZERO//)
      DO 52 I=1,N
      WRITE(6,200)(CSQ(I,J),J=1,N)
   52 CONTINUE
      WRITE(6,105)
  105 FORMAT(1H0,13H THE MATRIX E//)
      DO 502 I=1,N
      WRITE(6,200)(CSQ(I,J),J=NP1,L)
  502 CONTINUE
      WRITE(6,106)
  106 FORMAT(1H0,13H THE MATRIX P//)
      DO 503 I=NP1,L
      WRITE(6,200)(CSQ(I,J),J=1,N)
  503 CONTINUE
C
C     COMPUTATION OF R=PI E
C                       0
      DO 24 I=1,N
      DO 24 J=1,N
      B(I,J)=0.0D0
      DO 24 K=1,N
      J1=J+N
      B(I,J)=B(I,J)+CSQ(I,K)*CSQ(K,J1)
   24 CONTINUE
      DO 25 I=1,N
      DO 25 J=1,N
      CSQ(I,J)=0.0D0
      DO 25 K=1,N
      I1=I+N
      CSQ(I,J)=CSQ(I,J)+CSQ(I1,K)*B(K,J)
```

```
   25 CONTINUE
      WRITE(6,108)
  108 FORMAT(1H1,13H THE MATRIX R//)
      DO 530 I=1,N
      WRITE(6,200)(CSQ(I,J),J=1,N)
  530 CONTINUE
C                        +
C     COMPUTATION OF C =CRCRC
      DO 26 I=1,N
      DO 26 J=1,N
      J1=J+N
      B(I,J)=0.0D0
      DO 26 K=1,N
      K1=K+N
      B(I,J)=B(I,J)+CSQ(I,K)*CSQ(K1,J1)
   26 CONTINUE
      DO 27 I=1,N
      DO 27 J=1,N
      CSQ(I,J)=0.0D0
      DO 27 K=1,N
      CSQ(I,J)=CSQ(I,J)+B(I,K)*B(K,J)
   27 CONTINUE
      DO 28 I=1,N
      I1=I+N
      DO 28 J=1,N
      B(I,J)=0.0D0
      DO 28 K=1,N
      K1=K+N
      B(I,J)=B(I,J)+CSQ(I1,K1)*CSQ(K,J)
   28 CONTINUE
      WRITE(6,103)
  103 FORMAT(1H0,17H THE MATRIX CPLUS//)
      DO 53 I=1,N
      WRITE(6,200)(B(I,J),J=1,N)
   53 CONTINUE
C                        +   + *
C     COMPUTATION OF A =C A
      DO 29 I=1,N
      DO 29 J=1,M
      AP(I,J)=0.0D0
      DO 29 K=1,N
      AP(I,J)=AP(I,J)+B(I,K)*A(J,K)
   29 CONTINUE
      WRITE(6,104)
  104 FORMAT(1H0,17H THE MATRIX APLUS//)
      DO 54 I=1,N
      WRITE(6,200)(AP(I,J),J=1,M)
   54 CONTINUE
      RETURN
      END
```

THE MATRIX A

0.400000000000D 01	-0.100000000000D 01	-0.300000000000D 01	0.200000000000D 01
-0.200000000000D 01	0.500000000000D 01	-0.100000000000D 01	-0.300000000000D 01
0.200000000000D 01	0.300000000000D 01	-0.900000000000D 01	-0.500000000000D 01

THE MATRIX C

0.240000000000D 02	-0.800000000000D 01	-0.280000000000D 02	0.400000000000D 01
-0.800000000000D 01	0.350000000000D 02	-0.290000000000D 02	-0.320000000000D 02
-0.280000000000D 02	-0.290000000000D 02	0.910000000000D 02	0.420000000000D 02
0.400000000000D 01	-0.320000000000D 02	0.420000000000D 02	0.380000000000D 02

THE MATRIX CSQUARE

0.144000000000D 04	0.212000000000D 03	-0.282000000000D 04	-0.672000000000D 03
0.212000000000D 03	0.315400000000D 04	-0.477400000000D 04	-0.358600000000D 04
-0.282000000000D 04	-0.477400000000D 04	0.116700000000D 05	0.623400000000D 04
-0.672000000000D 03	-0.358600000000D 04	0.623400000000D 04	0.424800000000D 04

THE MATRIX IZERO

1.000000000000D 00	-0.277555756156D-16	0.555111512313D-16	-0.138777878078D-16
0.	1.000000000000D 00	-0.555111512313D-16	-0.555111512313D-16
-0.	0.	1.000000000000D 00	0.555111512313D-16
0.	0.	-0.	0.420996570938D-12

THE MATRIX E

0.	0.	0.856898029135D-04	0.
0.	0.832613684944D-03	0.340608203250D-03	0.
0.	0.350514832536D-01	-0.737268957518D-02	0.406440597361D-01
1.000000000000D 00	0.377475828373D-14	0.727272727273D 00	-0.909090909091D 00

THE MATRIX P

0.	0.	0.	1.000000000000D 00
0.	1.000000000000D 00	0.862401135151D 00	-0.649480469406D-14
1.000000000000D 00	0.409083119109D-00	-0.181396485072D-00	0.727272727273D 00
0.	0.	1.000000000000D 00	-0.909090909091D 00

THE MATRIX A*APLUS

1.000000000000D 00	-0.595357096955D-14	-0.274502642839D-13
0.582867087928D-14	1.000000000000D 00	0.693889390391D-14
-0.126565424807D-13	0.915933995316D-14	1.000000000000D 00

THE MATRIX A*APLUS*A

0.400000000000D 01	-1.000000000000D 00	-0.300000000000D 01	0.200000000000D 01
-0.200000000000D 01	0.500000000000D 01	-1.000000000000D 00	-0.300000000000D 01
0.200000000000D 01	0.300000000000D 01	-0.900000000000D 01	-0.500000000000D 01

THE MATRIX APLUS*A*APLUS

0.192280701754D-00	0.603508771930D-01	-0.364912280702D-01
0.200000000000D-00	0.300000000000D-00	-0.100000000000D-00
-0.561403508772D-02	0.529824561404D-01	-0.901754385965D-01
0.207017543860D-00	0.108771929825D-00	-0.112280701754D-00

THE MATRIX APLUS*A

0.575438596491D 00	-0.253408405371D-13	-0.308771929825D-00	0.385964912281D-00
0.643929354283D-14	1.000000000000D 00	0.320299342604D-13	0.328070903777D-13
-0.308771929825D-00	-0.752176099184D-14	0.775438596491D 00	0.280701754386D-00
0.385964912281D-00	-0.286992651866D-13	0.280701754386D-00	0.649122807018D 00

THE MATRIX R

0.420996570938D-12	0.158916029357D-26	0.306179324318D-12	-0.382724155398D-12
-0.763827139859D-17	0.310610526316D-01	-0.601760765550D-02	0.350514832536D-01
0.306132668328D-12	-0.601760765550D-02	0.156240684378D-02	-0.737268957546D-02
-0.382668644247D-12	0.350514832536D-01	-0.737268957546D-02	0.406440597365D-01

THE MATRIX CPLUS

0.419457063712D-01	0.602105263158D-01	0.540867959372D-02	0.504672206833D-01
0.602105263158D-01	0.140000000000D-00	0.237894736842D-01	0.852631578947D-01
0.540867959372D-02	0.237894736842D-01	0.109702677747D-01	0.147257617729D-01
0.504672206833D-01	0.852631578947D-01	0.147257617729D-01	0.672945521699D-01

THE MATRIX APLUS

0.192280701754D-00	0.603508771930D-01	-0.364912280702D-01
0.200000000000D-00	0.300000000000D-00	-0.100000000000D-00
-0.561403508772D-02	0.529824561404D-01	-0.901754385965D-01
0.207017543860D-00	0.108771929825D-00	-0.112280701754D-00

REFERENCES

(1) Autonne, L.: Sur les Matrices Hypohermitiennes et Sur les Matrices Unitaries. Ann. Ūniv. Lyon, (2) 38 (1915), 1-77.

(2) Bjerhammar, A.: Application of Calculus of Matrices to Method of Least Squares With Special Reference to Geodetic Calculations. Trans. Roy. Inst. Tech. Stockhold, 49(1951), 1-86.

(3) ______________: Rectangular Reciprocal Matrices With Special Reference to Geodetic Calculations. Bull. Geodesique, (1951), 188-220.

(4) Cecioni, F.: Sopra Operazioni Algebriche. Ann. Scu. Norm. Sup. Pisa, 11(1910), 17-20.

(5) Cline, R.: Note on the Generalized Inverse of the Product of Matrices, SIAM Review, 6(1964), 57-58.

(6) Cline, R.: Representations for the Generalized Inverse of Matrices With Applications in Linear Programming,Doctoral Thesis, Purdue University (1963).

(7) Decell, H.: An Explicit Form For the Generalized Inverse of an Arbitrary Complex Matrix, to appear.

(8) Greville, T. N. E.: The Pseudoinverse of a Rectangular or Singular Matrix and Its Applications to the Solution of Systems of Linear Equations. SIAM Review, 1 (1959), 38-43.

(9) ______________________: Some Applications of the Pseudo Inverse of a Matrix, SIAM Review, 2 (1960), 15-22.

(10) Moore, E. H.: Bull. Amer. Math. Soc., 26(1920), 394-395.

(11) Penrose, R.: A Generalized Inverse for Matrices. Proc. Camb. Philos. Soc., 51(1955), 406-413.

(12) ______________________: On Best Approximate Solution of Linear Matrix Equations. Proc. Camb. Philos. Soc., 52(1956), 17-19.